

Link Up Sync Up: Calculation of concordance and discordance rates between Independent Review Facility (IRF) and Investigator site data using SAS®

Girish Kankipati Seattle Genetics, Seattle, WA

Bhargav Koduru Seattle Genetics, Seattle, WA

ABSTRACT

Efficacy data, one of the important components of clinical trials, determines the effectiveness of investigational drug. In some phase II and III clinical trials response data is collected from both Independent Review Facility (IRF) and investigative sites. IRF verifies the endpoints in clinical trials by independent review of imaging data. Radiological assessment of images at the sites are often subjective, resulting in high interobserver variability among radiologists viewing the same images. To overcome this variability the images are also reviewed by an IRF. Central review by an IRF allows an auditable, blinded, and uniform process of evaluation. This provides greater consistency across sites, patients, and time points, achieving higher confidence levels in the data. Often IRF and site response data are assessed in clinical trials, and it is a best practice to check the consistency between these two sources of response.

This paper will explain the table and SAS® code for calculation of concordance rate and discordance rates for efficacy response rate endpoints between investigator assessment and IRF assessment using suitable examples and case scenarios. Concordance rate is the agreement or acceptance between IRF data and investigator data. Discordance rate is the inconsistency between these two assessments. This paper discusses various methodologies, challenges in calculating diagonal, off diagonal rates and confidence intervals by combining both types of data using Wilson, Agresti-Coull and Clopper-Pearson (Exact) methods. This analysis helps the sponsor to know the consistency across the data.

INTRODUCTION

Response data collected from investigator and IRF is mapped to Response (RS) SDTM domain. In oncology, response evaluations are generally collected under the following categories:

- Completed remission (CR)
- Partial Remission (PR)
- Stable Disease (SD)
- Progressive Disease (PD)
- Not Evaluable (NE).

In the table shell illustrated in Figure 1, correlation between two types of data is analyzed using statistical methods such as Wilson, Agresti-Coull and Clopper-Pearson (Exact). An example table shell shown in Figure 1, describes the calculation of concordance and discordance rates at End of Treatment visit for Overall Response. These rates can be calculated individually for each visit or certain visits can be chosen as per the study requirements. With this paper, we will show with an example code, how the concordance and discordance rates are calculated, along with their confidence intervals .

Investigator Assessment (or CRF):

Radiologists and oncologists at the clinical site review images and relevant clinical data to assess response. In a study, involving many clinical sites across multiple geographic regions there is a high likelihood of variation or inconsistency in assessment of response based on imaging data review at the clinical site.

Independent Review Facility (or IRF):

In order to address the probable variations introduced during Investigator assessment, radiological assessment of images done by a radiologist and/or oncologist independent from the clinical sites. Referred to as centralized independent review, it allows an auditable, blinded, and uniform process of evaluation.

Concordance rate:

A concordant pair is a pair in which the outcome is the same for each member of the pair. In this case, the pairs are the IRF and investigator assessments of outcome for a clinical trial subject. They are concordant when the IRF and investigator conclude the same response assessment for a subject. For example, both assessments are CR. The concordance rate is the number of subjects that are concordant over the total number of subjects assessed. As can be seen in Figure 1, the concordance rate is calculated by adding the diagonal counts and dividing by the total number of subjects assessed.

Discordance rate:

Discordance is the disagreement between IRF and investigator or CRF data. Discordance rate is the ratio between total number of discordant subjects to the total number of subjects assessed. As can be seen in Figure 1, the discordant rate is calculated by adding the off-diagonal counts and dividing by the total number of subjects assessed.

Figure 1 shows the table shell for the layout of concordant and discordant rates.

<Table x.x> Concordance of Overall Response at EOT by IRF and by Investigator mITT Analysis Set

IRF assessment	Investigator Assessment (N=XX) n (%)					
	CR	PR	SD	PD	NE	Total
	CR	x(xx)	x(xx)	x(xx)	x(xx)	x(xx)
PR	x(xx)	x(xx)	x(xx)	x(xx)	x(xx)	x(xx)
SD	x(xx)	x(xx)	x(xx)	x(xx)	x(xx)	x(xx)
PD	x(xx)	x(xx)	x(xx)	x(xx)	x(xx)	x(xx)
NE	x(xx)	x(xx)	x(xx)	x(xx)	x(xx)	x(xx)
Total	x(xx)	x(xx)	x(xx)	x(xx)	x(xx)	x(xx)
Concordant						x(xx)
Discordant						x(xx)

Figure 1. Shell for Concordance of Overall Response

Code for calculation of Concordance and Discordance Rates:

Create the format for corresponding AVAL values as per shell

```
proc format ;
value $bresp
    'COMPLETE METABOLIC RESPONSE','COMPLETE RESPONSE'           = 'CR'
    'PARTIAL METABOLIC RESPONSE','PARTIAL RESPONSE'               = 'PR'
    'NO METABOLIC RESPONSE','STABLE DISEASE'                       = 'SD'
    'PROGRESSIVE METABOLIC DISEASE','PROGRESSIVE DISEASE'         = 'PD'
    'NOT EVALUABLE'                                                = 'NE';

value $repnum
    'CR'='1'
    'PR'='2'
    'SD'='3'
    'PD'='4'
    'NE'='5'
    'Total'='6'
    'Concordant'='7'
    'Discordant'='8';

run;
```

Step-I:

Subset and combine CRF and IRF data.

- a) Filter the CRF & IRF data from response analysis dataset ADRS
- b) The dataset below is an example of ADRS, which consists of IRF and Investigator data. The variable AVAL represents the response data collected from both the sources. The variable RSEVAL indicates whether the assessment is from the IRF (Independent Assessor) or the Investigator. The following code subsets the data at EOT and creates a dataset with two new variables AVALIND and AVALINV for the IRF and investigator assessments, respectively.

STUDYID	USIBJID	SITEID	TRT1PN	PARCAT	PARCAT1
EFF	EFF-10001-10001	10001	1	LUGANO CLASSIFICATION	Metabolic Response
EFF	EFF-10001-10001	10001	1	LUGANO CLASSIFICATION	RADIOLOGIC PET AND CLINICAL ASSESSMENT
EFF	EFF-10002-10002	10002	2	LUGANO CLASSIFICATION	Metabolic Response
EFF	EFF-10002-10002	10002	2	LUGANO CLASSIFICATION	RADIOLOGIC PET AND CLINICAL ASSESSMENT

PARAM	PARAMCD	AVAL	AVALC	AVISIT	AVISITN	ADT	ADY	RSEVAL
Overall Respon..	OVLRESP	COMPLETE METABOLIC RESPONSE	COMPLETE ME..	End of Treatme..	999999	01JAN2018	40	INVESTIGATOR
Overall Respon..	OVLRESP	COMPLETE METABOLIC RESPONSE	COMPLETE ME..	End of Treatme..	999999	02JAN2018	57	INDEPENDENT ASSESSOR
Overall Respon..	OVLRESP	PARTIAL METABOLIC RESPONSE	PARTIAL META..	End of Treatme..	999999	03JAN2018	78	INVESTIGATOR
Overall Respon..	OVLRESP	NOT EVALUABLE	NOT EVALUAB..	End of Treatme..	999999	04JAN2018	65	INDEPENDENT ASSESSOR

Figure 2. Sample ADRS set with all the variables

```

proc sort data=adam.adrs out=irf (keep=usubjid aval rename=(aval=avalind))
nodupkey ; by usubjid ;
  where rseval='INDEPENDENT ASSESSOR' and avisit= "End of Treatment" and
paramcd='OVRLRESP' ;
run;

proc sort data=adam.adrs out=crf (keep=usubjid aval rename=(aval=avalinv) )
nodupkey;
  by usubjid ; where rseval='INVESTIGATOR' and avisit= "End of Treatment"
and paramcd='OVRLRESP' ;
run;

```

c) Merge CRF IRF and data sets and create flags with Y or N for concord

```

data slresp1(drop = avalind avalinv) ;
length irf inv $8 ;
  merge irf crf ;
  by usubjid ;

  irf = put(avalind,$bresp.) ;
  inv = put(avalinv,$bresp.) ;

  if irf ne '' and inv ne '' ;
  if irf ne '' and inv ne '' and irf = inv then concord = 'Y';
  else concord = 'N' ;
run;

data slresp2;
  length irf inv $8 ;
  set slresp1 ;
  if irf ne '' then irf = 'Total' ;
run;

data slresp3;
  length irf inv $8 ;
  set slresp1 ;
  if inv ne '' then inv = 'Total' ;
run;

```

	USUBJID	AVISIT	PARAMCD	AVALIND	IRF	AVALINV	INV	CONCORD
1	EFF-11001-0001	End of Treatment	OVRLRESP	COMPLETE METABOLIC RESPONSE	CR	COMPLETE METABOLIC RESPONSE	CR	Y
2	EFF-11002-0002	End of Treatment	OVRLRESP	PROGRESSIVE METABOLIC DISEASE	PD	PROGRESSIVE METABOLIC DISEASE	PD	Y
3	EFF-11002-0003	End of Treatment	OVRLRESP	COMPLETE METABOLIC RESPONSE	CR	COMPLETE METABOLIC RESPONSE	CR	Y
4	EFF-11002-0004	End of Treatment	OVRLRESP	COMPLETE METABOLIC RESPONSE	CR	COMPLETE METABOLIC RESPONSE	CR	Y
5	EFF-11002-0005	End of Treatment	OVRLRESP	NO METABOLIC RESPONSE	SD	NO METABOLIC RESPONSE	SD	Y

Figure 3. A Sample SAS® dataset after the merge of CRF and IRF data. Description of the variables is below.

AVALIND = AVAL values from IRF

AVALINV = AVAL values from Investigator

IRF = Formatted variable of AVALIND values.

INV = Formatted variable of AVALINV values.

CONCORD = Flag variable with Y or N values for Concordance.

Step-II:

Calculate the counts, percentages, and associated confidence intervals using Proc freq and generate the output.

```
/*-----*
|Calculate counts for CR PR SD PD & NE rows
*-----*/;
proc freq data=slresp1 order=freq ;
    tables irf*inv/out=conc nocol norow nopercent nocum;
run;

proc sort data=conc out=concl ;
    by irf;
run;

proc transpose data=concl out=concord ;
    id inv ;
    by irf;
    var count;
run;

/*-----*
|Calculate counts for total row
*-----*/;
data slresp2;
    length irf inv $8 ;
    set slresp1 ;
    if irf ne '' then irf = 'Total' ;
run;

proc freq data=slresp2 order=freq ;
    tables irf*inv/ out=con_tot nocol norow nopercent nocum;
run;

proc sort data=con_tot out=con_tot1 ;
    by irf;
run;

proc transpose data=con_tot1 out=con_tot2 ;
    id inv ;
    by irf;
    var count;
run;

/*-----*
|Count and create macro variable for BIGN
*-----*/;
proc sql ;
    select count (distinct usubjid) into :pop from slresp1 ;
quit;
```

```

/*-----*
|Calculate the percentages using BigN macro variable and impute the
|dummy zero values as per Statistician requirement or else leave blank
*-----*/;
data fin (rename=(cr_val =cr pr_val=pr sd_val=sd pd_val=pd ne_val=ne));
  length irf $200;
  set concord con_tot2;

  if cr ne . then
    cr_val=strip(put(cr,best.))||'('||strip(put(cr/&pop*100,5.2))||')';
  if pr ne . then
    pr_val=strip(put(pr,best.))||'('||strip(put(pr/&pop*100,5.2))||')';
  if sd ne . then
    sd_val=strip(put(sd,best.))||'('||strip(put(sd/&pop*100,5.2))||')';
  if pd ne . then
    pd_val=strip(put(pd,best.))||'('||strip(put(pd/&pop*100,5.2))||')';
  if ne ne . then
    ne_val=strip(put(ne,best.))||'('||strip(put(ne/&pop*100,5.2))||')';

  array terms cr_val pr_val sd_val pd_val ne_val;
  do over terms ;
    if terms =' ' then terms='0';
  end;

  keep irf cr_val pr_val sd_val pd_val ne_val;
run;

/*-----*
Calculate counts for Concordant, Discordant and Total with 'Y'
|or 'N' values
*-----*/;
data slresp2 ;
  set slresp2;
  output;
  concord='Z';
  output;
run;
proc freq data=slresp2 order=freq ;
  tables irf*concord/out=overall2 nocol norow nopercnt nocum;
run;

data tot_all (rename=(count=total));
  length irf $200;
  set overall overall2 ;

  if irf='Total' then do;
    if concord ='Y' then irf ='Concordant';
    else if concord ='N' then irf ='Discordant';
    else if concord ='Z' then irf ='Total';
  end;
  drop percent concord inv;
run;

proc sort data=fin ;
  by irf ;
run;

```

```

proc sort data=tot_all ;
  by irf ;
run;

data final(rename=(tot=total));
  merge fin(in=a) tot_all(in=b);
  by irf;
  if a or b;
  seq=input(put(irf, $REPNUM.),best.);
  tot=strip(put(total,best.));
  drop total;
run;

proc sort data=final ;
  by seq ;
run;

/*-----*
Calculate the Overall 90%-CI values for Concordant Discordant
subjects using Agresti-Coull, Clopper-Pearson (Exact) and Wilson
*-----*/;
proc sort data=slresp3 ;
  by concord inv;
run;

ods trace on ;
ods output BinomialCLs=Binomialcls;
proc freq data=slresp3 order=freq ;
  tables irf/ binomial(ac wilson exact) alpha=.1;
  by concord inv;
run;
ods trace off;

data Binomialcls ;
  set Binomialcls;
  if nmiss(lowercl, uppercl) = 0 then do;
    climit='('||trim(put((uppercl), 8.2 -1))||' , '||trim(put((lowercl),
      8.2 -1))||' )';
  end;
run;

proc sort data=Binomialcls out=Binomialcls1;
  by concord type;
run;

proc transpose data =Binomialcls1 out=trans_irf;
  id inv;
  var climit;
  by concord type;
run;

data conf_limit;
  length concord type irf $200 ;
  set trans_irf ;
  drop _name_;

```

```

if concord='Y' then irf='Concordant';
if concord='N' then irf='Discordant';
if irf='Concordant' then seq =11 ;
if irf='Discordant' then seq =12 ;
run;

data conc_disc;
  length total $100;
  set final conf_limit;
run;

options orientation=landscape;

title1 'Concordance of Overall Response at EOT per IRF and per Investigator'
justify=center;
title2 'Analysis Set' justify=center;

ods listing close;
ods rtf file="W:\Clinical Projects\abc\t-concord-mitt.rtf";
proc report data=conc_disc out=final2 split='|' nowd;
  column irf CR PR SD PD total seq;
  define seq/group noprint;
  define irf/ display 'IRF assessment' style(column)=[just=l]
width=700;
  define CR/display "CR" style(column)=[just=c] width=40;
  define PR/display "PR" style(column)=[just=c] width=40;
  define SD/display "SD" style(column)=[just=c] width=40;
  define PD/display "PD" style(column)=[just=c] width=40;
  define total/display "Total" style(column)=[just=c] width=40;
run;

```

The code generates the following table

**Concordance of Overall Response at EOT by IRF and by Investigator
Analysis Set**

IRF assessment	Investigator Assessment (N=148)					Total
	CR	PR	SD	PD	NE	
CR	88(59.46)	4(2.70)	0	0	0	92(62.16)
PR	8(5.41)	12(8.11)	4(2.70)	4(2.70)	0	28(18.92)
SD	0	0	12(8.11)	0	0	12(8.11)
PD	0	0	0	12(8.11)	0	12(8.11)
NE	0	0	0	0	4(2.70)	4(2.70)
Total	96(64.86)	16(10.81)	16(10.81)	16(10.81)	4(2.70)	148(100.0)
Concordant						128(86.49)
Discordant						20(13.51)
90% CI for Concordant						
Agresti-Coull						(0.77, 0.97)
Clopper-Pearson (Exact)						(0.75, 0.96)
Wilson						(0.77, 0.97)
90% CI for Discordant						
Agresti-Coull						(0.07, 0.19)
Clopper-Pearson (Exact)						(0.06, 0.18)
Wilson						(0.07, 0.19)

Figure 4. Output for Concordance of Overall Response at EOT by IRF and by Investigator table

Based on the above table we can observe that all diagonal rate values will add up to the total concordant value number i.e. $88+12+12+12+4=128$. The number of discordant pairs is calculated from the off-diagonal counts ($8+4+4+4=20$). .

Challenge:

Sometimes sites may miss the scans for EOT visits and may collect them at unscheduled visits. In this case, unscheduled scan visit data before EOT may be of clinical relevance. This should be clarified with the study medical monitor and biostatistician. If relevant, the following code can be used to identify unscheduled records from both IRF and CRF data and ensure that we are comparing data from similar visits.

```
/*-----*
|SQL code to identify subjects that do not have scans for EOT visit.
|Per SAP take unscheduled visits before EOT instead
*-----*/;
proc sql;
create table irf2(keep = usubjid avalind) as
  select unique
    usubjid, aval as avalind, avisit, avisitn, adt, paramcd, case
      when avisit= 'End Of Treatment' then 1
      else 0 end as eoty, max(calculated eoty) as eotfl
  from adam.adrs
  where rseval = 'INDEPENDENT ASSESSOR' and paramcd = 'OVRLRESP'
  group by usubjid having calculated eotfl = 0
  order by usubjid;
quit;

proc sql;
create table crf2(keep = usubjid avalinv) as
  select unique
    usubjid, aval as avalinv, avisit, avisitn, adt, paramcd, case
      when avisit= 'End Of Treatment' then 1
      else 0 end as eoty, max(calculated eoty) as eotfl
  from adam.adrs
  where rseval = 'INVESTIGATOR' and paramcd = 'OVRLRESP'
  group by usubjid
  having calculated eotfl = 0
  order by usubjid;
quit;
```

CONCLUSION

This paper presents calculation of concordance and discordance rates between investigator and IRF efficacy response assessments. Calculations of confidence intervals for the rates are also presented. Code for using unscheduled visits is provided for use when medical and statistical colleagues indicate that unscheduled visits are medically and statistically relevant.

REFERENCES

(n.d.). Retrieved from Statistics How To: <http://www.statisticshowto.com/probability-and-statistics/confidence-interval>

Applied Clinical Trials Editors. (2006, May 02). The Role of Independent Review in Oncology Trials. *Applied Clinical Trials*. Retrieved from <http://www.appliedclinicaltrials.com/role-independent-review-oncology-trials>

Ford, R. R., Ford, R. W., Neal, M. O., Kahl, B. S., Chen, L., Munteanu, M., & Cheson, B. D. (2017). Investigator and independent review committee exploratory assessment and verification of tumor response in a non-Hodgkin lymphoma study. *Full Terms & Conditions of access and use can be found at*, 1332-1340. doi:10.1080/10428194.2016.1233535

ACKNOWLEDGMENTS

We would like to acknowledge Avani Kaja for their valuable contributions for this paper. We also would like sincerely thank to John Shaik, Nancy Yuan and Shefalica Chand for their valuable suggestions.

CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the author at:

Girish Kankipati
Seattle Genetics, Inc.
21823 - 30th Drive S.E.
Bothell, WA 98021
425-527-2140
gkankipati@seagen.com

Bhargav Koduru
Seattle Genetics, Inc.
21823 - 30th Drive S.E.
Bothell, WA 98021
425-527-4715
bkoduru@seagen.com

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration. Other brand and product names are trademarks of their respective companies.